cmake_minimum_required(VERSION 3.16)

project(BiliFormat VERSION 0.1 LANGUAGES CXX)
include(FetchContent)

set(CMAKE_AUTOUIC ON)
set(CMAKE_AUTOMOC ON)
set(CMAKE_AUTORCC ON)

set(CMAKE_CXX_STANDARD 17)
set(CMAKE_CXX_STANDARD_REQUIRED ON)

find_package(QT NAMES Qt6 Qt5 REQUIRED COMPONENTS Widgets)
find_package(Qt${QT_VERSION_MAJOR} REQUIRED COMPONENTS Widgets)

set(PROJECT_SOURCES
        src/main.cc
        src/mainwindow.cc
        src/mainwindow.h
        src/mainwindow.ui
        src/scan.cc
        src/videoinfo.cc
)

message(STATUS "Build BiliFormat: " ${PROJECT_VERSION})
if (${QT_VERSION_MAJOR} GREATER_EQUAL 6)
    qt_add_executable(BiliFormat
            MANUAL_FINALIZATION
            ${PROJECT_SOURCES}
    )
    # Define target properties for Android with Qt 6 as:
    #    set_property(TARGET BiliFormat APPEND PROPERTY QT_ANDROID_PACKAGE_SOURCE_DIR
    #                 ${CMAKE_CURRENT_SOURCE_DIR}/android)
    # For more information, see https://doc.qt.io/qt-6/qt-add-executable.html#target-creation
else ()
    if (ANDROID)
        add_library(BiliFormat SHARED
                ${PROJECT_SOURCES}
        )
        # Define properties for Android with Qt 5 after find_package() calls as:
        #    set(ANDROID_PACKAGE_SOURCE_DIR "${CMAKE_CURRENT_SOURCE_DIR}/android")
    else ()
        add_executable(BiliFormat
                ${PROJECT_SOURCES}
        )
    endif ()
endif ()

target_link_libraries(BiliFormat PRIVATE Qt${QT_VERSION_MAJOR}::Widgets)
target_link_directories(BiliFormat PRIVATE ${CMAKE_CURRENT_SOURCE_DIR}/src/)

# ffmpeg
target_include_directories(BiliFormat PRIVATE $ENV{ffmpeg_dir}/include)
target_link_directories(BiliFormat PRIVATE $ENV{ffmpeg_dir}/lib)
target_link_libraries(BiliFormat PRIVATE avcodec)
target_link_libraries(BiliFormat PRIVATE avdevice)
target_link_libraries(BiliFormat PRIVATE avformat)
target_link_libraries(BiliFormat PRIVATE swscale)
target_link_libraries(BiliFormat PRIVATE avutil)

# nlohmann_json
FetchContent_Declare(
        json
        URL https://github.com/nlohmann/json/releases/download/v3.11.3/json.tar.xz
        DOWNLOAD_EXTRACT_TIMESTAMP TRUE
)
FetchContent_MakeAvailable(json)
target_link_libraries(BiliFormat PRIVATE nlohmann_json::nlohmann_json)

# spdlog
FetchContent_Declare(
        spdlog
        URL https://github.com/gabime/spdlog/archive/refs/tags/v1.14.1.zip
        DOWNLOAD_EXTRACT_TIMESTAMP TRUE
)
FetchContent_MakeAvailable(spdlog)
target_link_libraries(BiliFormat PRIVATE spdlog::spdlog)

# build
if (${QT_VERSION} VERSION_LESS 6.1.0)
    set(BUNDLE_ID_OPTION MACOSX_BUNDLE_GUI_IDENTIFIER com.example.BiliFormat)
endif ()
set_target_properties(BiliFormat PROPERTIES
        ${BUNDLE_ID_OPTION}
        MACOSX_BUNDLE_BUNDLE_VERSION ${PROJECT_VERSION}
        MACOSX_BUNDLE_SHORT_VERSION_STRING ${PROJECT_VERSION_MAJOR}.${PROJECT_VERSION_MINOR}
        MACOSX_BUNDLE TRUE
        WIN32_EXECUTABLE TRUE
)

# install
include(GNUInstallDirs)
install(TARGETS BiliFormat
        BUNDLE DESTINATION .
        LIBRARY DESTINATION ${CMAKE_INSTALL_LIBDIR}
        RUNTIME DESTINATION ${CMAKE_INSTALL_BINDIR}
)

if (QT_VERSION_MAJOR EQUAL 6)
    qt_finalize_executable(BiliFormat)
endif ()


# test
option(INSTALL_GTEST "Enable installation of googletest." OFF)
FetchContent_Declare(
        googletest
        URL https://github.com/google/googletest/archive/refs/tags/v1.15.0.zip
        DOWNLOAD_EXTRACT_TIMESTAMP TRUE
)
FetchContent_MakeAvailable(googletest)
enable_testing()
add_subdirectory(${CMAKE_SOURCE_DIR}/tests)
